Computers & Geosciences Vol. 7, No. 4, pp. 401–405, 1981
Printed in Great Britain.

0098-3004/81/040401-05$02.00/0

SHORT NOTES

AN ALGORITHM FOR SYNTHESIZING THE GEOMAGNETIC FIELD

S. R. C. Malin and D. R. Barraclough
Geomagnetism Unit, Institute of Geological Sciences, West Mains Road, Edinburgh EH9 3LA, U.K.

(*Received* 19 *December* 1980)

INTRODUCTION

Leaving aside transient and periodic variations, the Earth's magnetic field may be considered to be the sum of two parts: the main geomagnetic field and the anomaly field. The main field originates in the Earth's fluid core and shows only large-scale features at the surface of the Earth. The anomaly field has its sources in the crust and has features which are typically 1000 km or less in size.

Knowledge of the main geomagnetic field and its annual rate of change (secular variation) is required for many purposes. For example, in the study of local anomalies, it is necessary to remove the regional trend (Bullard, 1967). If the trend surface is based on a global model rather than a local polynomial, there will be no discontinuities between anomaly charts for adjacent regions. Main-field values are required also in the calculation of trajectories of charged particles in the magnetosphere and ionosphere (Akasofu and Chapman, 1972). Large scale maps of the magnetic declination (difference between true north and magnetic north) derived from models of the main field continue to be used widely for navigational purposes.

It was with these facts in mind that the International Association of Geomagnetism and Aeronomy (IAGA) adopted an International Geomagnetic Reference Field (IGRF). It was, however, recognized that this model would not be adequate for navigational use (Zmuda, 1971). The original IGRF (IAGA, 1969) consisted of spherical harmonic coefficients for the main field at 1965.0 and for its secular variation, both sets of coefficients extending to eighth order and degree. By the mid-1970's, the model was no longer an accurate description, largely because of changes in the secular variation and IAGA (1975) adopted a revised version which we denote IGRF (1975). To preserve continuity in the main-field values, the revision was limited to the addition of a new set of 80 coefficients describing the secular variation for the period 1975 to 1980. IGRF (1975) thus consists of three sets of 80 spherical harmonic coefficients: one for the main-field at 1975.0 (the 1965.0 main field coefficients corrected to 1975.0 using the original secular variation coefficients), the original secular variation coefficients to be used for deriving field values for dates before 1975 and the new secular variation coefficients to be used for dates after 1975.

Subroutines and programs for the evaluation of the original IGRF have been published previously, though not in readily-accessible form (Cain and others, 1968; Fabiano and Peddie, 1969; Malin, 1969; Kluge, 1970a, b; Barraclough and Malin, 1971; Stassinopoulos and Mead, 1972); they are available also from the World Data Centers. Here we present a FORTRAN subroutine (see Appendix) which computes geomagnetic field values for any specified date and point in space from the IGRF coefficients.

In practice, the synthesized field values will be realistic only in the region from the surface of the Earth's core (radius = 3500 km, approximately) out to a distance of 4 Earth radii. The temporal range of coefficients is from 1955.0 to the present (Zmuda, 1971, p. 186). To ensure that the subroutine is compatible with most computers, it is written deliberately in unsophisticated FORTRAN using a minimum of storage space, arrays of one dimension only and a limited range of statement types and library subroutines.

METHOD OF COMPUTATION

Before computing the geomagnetic field components, we must correct the spherical harmonic coefficients to the required date (DATE, specified in years, A.D.), using the relations

$$g_n^m = g_n^m(1975) + t\dot{g}_n^m(<1975),\quad h_n^m = h_n^m(1975) + t\dot{h}_n^m(<1975);\ (t<0),$$

$$g_n^m = g_n^m(1975) + t\dot{g}_n^m(>1975),\quad h_n^m = h_n^m(1975) + t\dot{h}_n^m(>1975);\ (t\geq 0),$$

where $t = \text{DATE} - 1975.0$. Here, $g_n^m(1975)$, $h_n^m(1975)$ denote the IGRF main-field coefficients for 1975; $\dot{g}_n^m(<1975)$, $\dot{h}_n^m(<1975)$ denote the IGRF secular-variation coefficients for use when DATE < 1975 and $\dot{g}_n^m(>1975)$, $\dot{h}_n^m(>1975)$ denote those for use when DATE > 1975.

The north (X), east (Y), and vertically downward (Z) components of the geomagnetic field are given by

$$X = \sum_{n=1}^{8} \left\{ g_n^0(\bar{a}/r)^{n+2}.X_n^0 + \sum_{m=1}^{n} [(g_n^m \cos m\lambda + h_n^m \sin m\lambda)(\bar{a}/r)^{n+2}.X_n^m] \right\},$$

$$Y = \sum_{n=1}^{8} \sum_{m=1}^{n} [(g_n^m \sin m\lambda - h_n^m \cos m\lambda)(\bar{a}/r)^{n+2}.\ Y_n^m],$$

$$Z = -\sum_{n=1}^{8} \left\{ g_n^0 (\bar{a}/r)^{n+2}.\ Z_n^0 + \sum_{m=1}^{n} [(g_n^m \cos m\lambda + h_n^m \sin m\lambda)(\bar{a}/r)^{n+2}.\ Z_n^m] \right\},$$

where $\bar{a}$ denotes the radius of the reference sphere, r denotes the radial distance from the center of the Earth, λ denotes east longitude, m and n denote integers and X_n^m, Y_n^m, Z_n^m are defined in terms of the associated Legendre polynomial P_n^m of degree n and order m. The Schmidt quasinormalized form of the associated Legendre polynomial is defined by:

$$P_n^m(x) = \frac{1}{2^n.n!} \left[\frac{(2-\delta_{0m})(n-m)!(1-x^2)^m}{(n+m)!} \right]^{1/2} \frac{d^{m+n}}{dx^{m+n}}(x^2-1)^n,$$

where δ_{ij} is the Kronecker delta ($\delta_{ii} = 1$; $\delta_{ij} = 0$ for $i \neq j$) and $x = \cos\theta$, where θ denotes geocentric colatitude. Finally, $X_n^m = dP_n^m/d\theta$, $Y_n^m = mP_n^m/\sin\theta$ (except at the poles where $Y_n^m = X_n^m \cos\theta$) and $Z_n^m = (n+1)P_n^m$. Note that definitions of X_n^m, Y_n^m used here differ by a factor n from those frequently used elsewhere.

The IGRF coefficients are referred to a sphere of radius $\bar{a} = 6371.2$ km, whereas the Earth's surface corresponds closely to an ellipsoid. Thus, if a point is defined relative to the Earth's surface, that is in geodetic coordinates θ' = colatitude, λ = east longitude and h = orthometric height above mean sea level, these must be converted first to geocentric coordinates θ, λ and r. This conversion is performed by the subroutine if ITYPE is specified as 1. The International Astronomical Union ellipsoid, which has an equatorial radius, a, of 6378.16 km and a reciprocal flattening, $1/f$, of 298.25, was recommended for use with the IGRF and is employed in this subroutine; h is taken to be the height above this ellipsoid measured along the normal. Alternatively, the point may be given directly in geocentric coordinates, in which case ITYPE = 2.

Conversion from geodetic to geocentric coordinates is as follows:

$$r = [h(h+2) + (a^4 \sin^2\theta' + b^4 \cos^2\theta')/\rho^2]^{1/2},$$

where

$$\rho = (a^2 \sin^2\theta' + b^2 \cos^2\theta')^{1/2};$$

$$\cos\delta = (h+\rho)/r;$$

$$\sin\delta = (a^2 - b^2).\cos\theta'.\sin\theta'/\rho r;$$

$$\cos\theta = \cos\theta'.\cos\delta - \sin\theta'.\sin\delta;$$

$$\sin\theta = \sin\theta'.\cos\delta + \cos\theta'.\sin\delta;$$

here a denotes the equatorial radius and b denotes the polar radius ($=a[1-f]$), and $\delta = \theta - \theta'$. The quantities $\cos\delta$ and $\sin\delta$ are used later to convert X and Z back to the geodetic system of coordinates, if this is specified by the input data (when ITYPE = 1). If geocentric coordinates are employed, $\delta = 0$.

Four values of P_n^m and X_n^m are specified:

$$P_0^0 = 1, X_0^0 = 0, P_1^1 = \sin\theta, X_1^1 = \cos\theta$$

and the remaining values are calculated using the recurrence relationships:

$$P_n^n = (1 - 1/2n)^{1/2} \sin\theta.\ P_{n-1}^{n-1},$$

$$X_n^n = (1 - 1/2n)^{1/2} (\sin\theta.\ X_{n-1}^{n-1} + \cos\theta.\ P_{n-1}^{n-1}),$$

$$P_n^m = \{(2n-1)\cos\theta.\ P_{n-1}^m - [(n-1)^2 - m^2]^{1/2} P_{n-2}^m\}/(n^2 - m^2)^{1/2},$$

$$X_n^m = \{(2n-1)[\cos\theta.\ X_{n-1}^m - \sin\theta.\ P_{n-1}^m] - [(n-1)^2 - m^2]^{1/2} X_{n-2}^m\}/(n^2 - m^2)^{1/2}.$$

At the same time, $\cos m\lambda$ and $\sin m\lambda$ are computed by

$$\cos m\lambda = \cos(m-1)\lambda.\cos\lambda - \sin(m-1)\lambda.\sin\lambda,$$

$$\sin m\lambda = \sin(m-1)\lambda.\cos\lambda + \cos(m-1)\lambda.\sin\lambda.$$

Finally, X and Z are adjusted to the appropriate coordinate system and F, the magnitude of the total field, is computed:

$$X_{geodetic} = X\cos\delta + Z\sin\delta,$$

$$Z_{geodetic} = Z\cos\delta - X\sin\delta,$$

and

$$F = (X^2 + Y^2 + Z^2)^{1/2}.$$

The remaining geomagnetic elements may be calculated from those supplied by the subroutine as follows:

$$H = (X^2 + Y^2)^{1/2}; D = \arctan(Y/X); I = \arctan(Z/H),$$

where H denotes the horizontal component of the field; D denotes the easterly declination of the field from geographical north and I denotes the inclination of the field to the horizontal.

The subroutine is called by the statement:

CALL MAGSYN (DATE, ITYPE, ALT, COLAT, ELONG, X, Y, Z, F).

X, Y, Z and F are supplied by the subroutine; the other values in the CALL statement remain unaltered.

Much of the subroutine is explained in comment statements or is self-evident. However, the following remarks may help to clarify the procedure. The main-field coefficients are stored in the array GH(L), L = 1 to 80 where $L = n^2 + 2m$ for g_n^0 and h_n^m; $L = n^2 + 2m - 1$ for g_n^m ($m \neq 0$). The pre-1975 secular variation coefficients are similarly stored in the array SV(L), L = 1 to 80 and the post-1975 coefficients in SV(80 + L), L = 1 to 80. The quantities P_n^m and X_n^m are stored in the arrays P(K) and Q(K), K = 1 to 45, where $K = n(n+1)/2 + m + 1$, and starting with P_0^0, X_0^0. Intermediate floating-point quantities are stored in ONE, TWO, and THREE; integer quantities I, J and LL. The FORTRAN variables which correspond to those employed in the previous formulae are:

$$\mathrm{ALT} \equiv h; \mathrm{COLAT} \equiv \theta' \text{ (if ITYPE = 1)};$$

$$\mathrm{ALT} \equiv r; \mathrm{COLAT} \equiv \theta \text{ (if ITYPE = 2)};$$

$$\mathrm{ELONG} \equiv \lambda;\ \mathrm{CL(M)} \equiv \cos m\lambda; \mathrm{SL(M)} \equiv \sin m\lambda;$$

$$\mathrm{T} \equiv t; \mathrm{R} \equiv r; \mathrm{CT} \equiv \cos\theta; \mathrm{ST} \equiv \sin\theta;$$

$$\mathrm{CD} \equiv \cos\delta; \mathrm{SD} \equiv \sin\delta; \mathrm{A2} \equiv a^2; \mathrm{B2} \equiv b^2; \mathrm{RHO} \equiv \rho; \mathrm{RATIO} \equiv \bar{a}/r;$$

$$\mathrm{RR} \equiv (\bar{a}/r)^{n+2}; \mathrm{FM \text{ and } M} \equiv m; \mathrm{FN \text{ and } N} \equiv n;$$

$$\mathrm{GM} \equiv m^2; \mathrm{GN} \equiv n-1.$$

MODIFICATIONS REQUIRED FOR USE WITH OTHER MODELS

The subroutine is written specifically for the synthesis of IGRF(1975) values. However, it may be modified readily for the synthesis of field values from other spherical harmonic models by altering the statements marked by two asterisks.

The numerical values of the coefficients in the DATA statements must be changed to those of the required model. Most models have only one set of secular variation coefficients; these should be stored in the array SV and the statement

$$\mathrm{IF\,(T.GE.0.0)\,LL} = 80$$

should be deleted. If the maximum order and degree, n^*, differs from 8, the numbers in the DIMENSION statement must be changed from 80 or 160 to $n^*(n^*+2)$; from 45 to $(n^*+1)(n^*+2)/2$ and from 8 to n^*. The final value in the DO statement must be changed also from 45 to $(n^*+1)(n^*+2)/2$. If the date to which the main-field coefficients refer differs from 1975.0, the appropriate date must be substituted for 1975.0 in the first assignment statement.

The listing of the subroutine includes, in comment statements near the beginning, details of the input and output parameters used in calling the subroutine. Further comment statements at the end of the listing give numerical examples. In these, the numbers represent values of the parameters in the same sequence as in the CALL statement. For example, a CALL of MAGSYN with DATE = 1979.2, ITYPE = 1, ALT = 0.0 km, COLAT = 40°.0 (latitude 50°N) and ELONG = 5°.0 should give the following results: X = 19778 nT, Y = −1330 nT, Z = 43408 nT, F = 47720 nT.

Acknowledgments—We are grateful to Mr. B. R. Leaton for helpful suggestions. This paper is published with the permission of the director of the Institute of Geological Sciences (N.E.R.C.).

REFERENCES

Akasofu, S.-I., and Chapman, S., 1972, Solar-terrestrial physics: Clarendon Press, Oxford, 901 p.

Barraclough, D.R. and Malin, S.R.C., 1971, Synthesis of International Geomagnetic Reference Field values: Inst. Geol. Sci. Report, No. 71/1, 26 p.

Bullard, E. C., 1967, The removal of trend from magnetic surveys: Earth planet. Sci. Letters, v2, p. 293–300.

Cain, J. C., Hendricks, S., Daniels, W. and Jensen, D. C., 1968, Computation of the main geomagnetic field from spherical harmonic expansions: Natl. Space Sci. Data Cent., Data Users' Note, NSSDC 68-11, 46 p.

Fabiano, E. B., and Peddie, N. W., 1969, Grid values of the total magnetic intensity IGRF-1965, Environ. Sci. Serv. Adm., Tech. Report, C & GS 38, 55p.

IAGA Commission 2 Working Group 4, 1969, International Geomagnetic Reference Field 1965.0, Jour. Geophys. Res., v. 74, no. 17, p 4407–4408; Jour. Geomagn. Geoelect., Kyoto, v. 21, no. 2, 569; Geomagn. Aeron., v. 9, no. 5, p. 956.

IAGA Division I Study Group on Geomagnetic Reference Fields, 1975, International Geomagnetic Reference Field 1975, Jour. Geomagn. Geoelect., Kyoto, v. 27, no. 5, p 437–439. Also: Geophys. Jour. Roy Astro. Soc., v. 44, p 733–734 (1976); EOS Am. Geophys. Union Trans. v. 57, no. 3, p 120–121 (1976); Jour. Geophys. Res., v. 81, no. 28, p. 5163–5164 (1976); Geomagn. Aeron., v. 17, no. 1, p. 167–168 (1977).

Kluge, G., 1970a, A generalised method for the calculation of the geomagnetic field from multipole expansions: Eur. Space Oper. Cent., Internal Note, No. 61, 27 p.

Kluge, G., 1970b, Computer program SHELL for the calculation of B and L from models of the geomagnetic field: Eur. Space Oper. Cent., Internal Note, No. 67, 17 p.

Malin, S. R. C., 1969, Synthesis of International Geomagnetic Reference Field values: Inst. Geol. Sci. Geomagn. Unit Report, No. 2, 7 p.

Stassinopoulos, E. G., and Mead, G. D., 1972, ALLMAG, GDALMG, LINTRA: Computer programs for geomagnetic field and field-line calculations: Natl. Space Sci. Data Cent. Report, NSSDA 72-12, 51p.

Zmuda, A. J., 1971, World Magnetic Survey 1957-1969: Bull. Int. Assoc. Geogmagn. Aeron., No. 28, p. 148–152.

APPENDIX

Program listing

```
      SUBROUTINE MAGSYN (DATE,ITYPE,ALT,COLAT,ELONG,X,Y,Z,F)
C
C     THIS SUBROUTINE COMPUTES VALUES OF X,Y,Z AND F FOR A GIVEN DATE
C     AND POSITION FROM THE SPHERICAL HARMONIC COEFFICIENTS OF THE
C     INTERNATIONAL GEOMAGNETIC REFERENCE FIELD.
C
C     DATA   DATE = REQUIRED DATE IN YEARS AND DECIMALS OF A YEAR, A.D.
C           ITYPE = 1 IF GEODETIC   COORDINATES ARE BEING USED
C           ITYPE = 2 IF GEOCENTRIC COORDINATES ARE BEING USED
C             ALT = HEIGHT ABOVE MEAN SEA LEVEL IN KM          (ITYPE=1)
C             ALT = RADIAL DISTANCE FROM CENTRE OF EARTH IN KM (ITYPE=2)
C           COLAT = COLATITUDE      IN DEGREES (0.0 TO 180.0)
C           ELONG = EAST LONGITUDE IN DEGREES (0.0 TO 360.0)
```

```
C
C     OUTPUT    X = NORTH    COMPONENT OF MAGNETIC FORCE IN NT
C               Y = EAST     COMPONENT OF MAGNETIC FORCE IN NT
C               Z = VERTICAL COMPONENT OF MAGNETIC FORCE IN NT
C                                             (POSITIVE DOWNWARDS).
C               F = TOTAL MAGNETIC FORCE IN NT
C     N.B. THE COORDINATE SYSTEM FOR X,Y AND Z IS THE SAME AS THAT
C          SPECIFIED BY ITYPE.
C
      DIMENSION GH(80),SV(160),P(45),Q(45),CL(8),SL(8)                  **
C
C     THE S.H.COEFFICIENTS ARE SUPPLIED BY THE FOLLOWING DATA STATEMENTS
C     (OR THE COEFFICIENTS MAY BE READ IN THE MAIN PROGRAM AND MADE
C     AVAILABLE TO THE SUBROUTINE BY A COMMON STATEMENT.)
C
      DATA GH/      -30186.,-2036.,5735.,-1898.,2997.,-2124.,1551.,-37.,
     1         1299.,-2144., -361.,1296.,  249., 805., -253., 951.,807.,
     2          148.,  462., -264.,-393.,   37., 235., -307.,-204.,368.,
     3           39.,  275.,  142., -20., -147.,-161.,  -99., -38., 74.,
     4           46.,   57.,  -23.,  15.,  102.,-210.,   88.,  -1.,-43.,
     5           -8.,   -9., -114.,  -4.,   66., -57.,  -68.,  -7.,-24.,
     6            7.,   -4.,  -22.,  11.,   -9.,  27.,   11., -17., -8.,
     7          -14.,   11.,   13.,   4.,    3., -15.,  -12.,   2., -4.,
     8          -19.,    6.,    1.,  -2.,   18.,   9.,   -6.,   1.,-19./
      DATA SV/   15.3, 8.7, -2.3,-24.4, 0.3,-11.8,-1.6,-16.7, 0.2,-10.8,
     1            4.2, 0.7,  0.7, -3.8,-7.7, -0.7, 0.2, -0.1,-3.0,  1.6,
     2           -0.1, 2.9, -2.1, -4.2, 1.9,  1.1, 2.3,  2.9, 1.7,  0.6,
     3           -2.4, 0.0,  0.8,  1.3,-0.3, -0.1,-0.3, -0.9, 1.1, -0.4,
     4            1.9, 2.0, -0.4, -1.1,-0.4,  0.1,-0.2,  0.9,-0.5, -0.3,
     5           -1.1,-0.7,  0.3, -0.5, 0.4,  0.3, 0.2,  0.0, 0.4, -0.2,
     6            0.2,-0.6,  0.3,  0.1, 0.4,  0.1, 0.6, -0.2, 0.0, -0.3,
     7            0.0,-0.2, -0.1, -0.3, 0.3, -0.4,-0.3, -0.3,-0.5, -0.3,
     8           25.6,10.0,-10.2,-24.9, 0.7, -3.0, 4.3,-18.9,-3.8,-10.4,
     9            6.9,-4.1,  2.5, -4.2,-5.0, -0.2,-2.0,  5.0,-3.9,  0.8,
     A           -2.1, 1.7, -3.1, -1.0, 0.3, -0.7, 1.2,  1.1, 2.3, -1.6,
     B           -2.0,-0.5,  1.3,  1.0, 1.1,  0.2, 0.5, -0.5, 2.0, -0.1,
     C            2.8,-0.2,  0.0, -1.3, 0.9,  0.7,-0.1,  1.7, 0.0,  0.0,
     D           -1.4, 0.0, -0.1,  0.6, 0.3,  0.9, 0.3,  0.3,-0.7,  0.3,
     E            0.1,-0.5,  0.8,  0.2, 0.3, -0.2, 0.0, -0.4, 0.2, -0.2,
     F           -0.4,-0.3, -0.3,  0.4, 0.6, -0.3,-0.3, -0.6,-0.1,  0.3/
C
C     SET INITIAL VALUES
C
      T     = DATE - 1975.0                                             **
      R     = ALT
      ONE   = COLAT*0.0174533
      CT    = COS(ONE)
      ST    = SIN(ONE)
      ONE   = ELONG*0.0174533
      CL(1) = COS(ONE)
      SL(1) = SIN(ONE)
      X     = 0.0
      Y     = 0.0
      Z     = 0.0
      CD    = 1.0
      SD    = 0.0
      L     = 1
      LL    = 0
      IF (T.GE.0.0) LL = 80                                             **
      M     = 1
      N     = 0
      IF (ITYPE.EQ.2) GO TO 1
C
C     CONVERSION FROM GEODETIC TO GEOCENTRIC COORDINATES
C
      A2    = 40680925.
      B2    = 40408585.
      ONE   = A2*ST*ST
      TWO   = B2*CT*CT
      THREE = ONE + TWO
      RHO   = SQRT(THREE)
      R     = SQRT(ALT*(ALT + 2.0*RHO) + (A2*ONE + B2*TWO)/THREE)
      CD    = (ALT + RHO)/R
      SD    = (A2 - B2)/RHO*CT*ST/R
      ONE   = CT
      CT    = CT*CD -  ST*SD
      ST    = ST*CD + ONE*SD
C
    1 RATIO = 6371.2/R
      RR    = RATIO*RATIO
```

```
C
C     COMPUTATION OF SCHMIDT QUASI-NORMAL COEFFICIENTS P AND X(=Q)
C
      P(1)  = 1.0
      P(3)  = ST
      Q(1)  = 0.0
      Q(3)  =  CT
      DO 8 K=2,45                                                        **
      IF (N.GE.M) GO TO 2
      M     = 0
      N     = N + 1
      RR    = RR*RATIO
      FN    = N
      GN    = N - 1
    2 FM    = M
      IF (M.NE.N) GO TO 3
      IF (K.EQ.3) GO TO 4
      ONE   = SQRT(1.0 - 0.5/FM)
      J     = K - N - 1
      P(K)  = ONE*ST*P(J)
      Q(K)  = ONE*(ST*Q(J) + CT*P(J))
      CL(M) = CL(M-1)*CL(1) - SL(M-1)*SL(1)
      SL(M) = SL(M-1)*CL(1) + CL(M-1)*SL(1)
      GO TO 4
    3 GM    = M*M
      ONE   = SQRT(FN*FN - GM)
      TWO   = SQRT(GN*GN - GM)/ONE
      THREE = (FN + GN)/ONE
      I     = K - N
      J     = I - N + 1
      P(K)  = THREE*CT*P(I) - TWO*P(J)
      Q(K)  = THREE*(CT*Q(I) - ST*P(I)) - TWO*Q(J)
C
C     SYNTHESIS OF X, Y AND Z IN GEOCENTRIC COORDINATES
C
    4 ONE   = (GH(L) + SV(LL+L)*T)*RR
      IF (M.EQ.0) GO TO 7
      TWO   = (GH(L+1) + SV(LL+L+1)*T)*RR
      THREE = ONE*CL(M) + TWO*SL(M)
      X     = X + THREE*Q(K)
      Z     = Z - (FN + 1.0)*THREE*P(K)
      IF (ST.EQ.0.0) GO TO 5
      Y     = Y + (ONE*SL(M) - TWO*CL(M))*FM*P(K)/ST
      GO TO 6
    5 Y     = Y + (ONE*SL(M) - TWO*CL(M))*Q(K)*CT
    6 L     = L + 2
      GO TO 8
    7 X     = X + ONE*Q(K)
      Z     = Z - (FN + 1.0)*ONE*P(K)
      L     = L + 1
    8 M     = M + 1
C
C     CONVERSION TO COORDINATE SYSTEM SPECIFIED BY ITYPE
C
      ONE   = X
      X     = X*CD +   Z*SD
      Z     = Z*CD - ONE*SD
      F     = SQRT(X*X + Y*Y + Z*Z)
C
      RETURN
      END
C
C     LIBRARY FUNCTIONS USED BY THIS SUBROUTINE ARE:   SIN,COS,SQRT
C
C     SAMPLE RESULTS:
C     MAGSYN(1965.0,1,   0.0,  0.0, 30.0,  2541., -240., 56349.,56407.)
C     MAGSYN(1969.3,1,   0.0, 57.0,195.0, 25092., 5729., 30145.,39638.)
C     MAGSYN(1965.0,2,6371.2, 45.0, 10.0, 21848.,-1052., 40417.,45956.)
C     MAGSYN(1967.5,2,9000.0, 31.0,359.0,  5750.,-1226., 17853.,18796.)
C     MAGSYN(1975.0,1,   0.0, 57.0,195.0, 25072., 5721., 30120.,39605.)
C     MAGSYN(1979.2,1,   0.0, 40.0,  5.0, 19778.,-1330., 43408.,47720.)
```